

檢視參照與資料庫函數

10-1

多重條件判斷IFS()

```
IFS(邏輯測試1,成立值1,[邏輯測試2,成立值2],[邏輯測試3,成立值3], ...)
IFS(logical_test1,value_if_true1,[logical_test2,value_if_true2],[logical_test3,
value_if_true3], ...)
```

邏輯測試1、邏輯測試2、…，為可以產生TRUE或FALSE結果的任何條件式，最多可擁有127組測試。若邏輯測試1條件式成立，即取成立值1之運算結果；反之，繼續判斷邏輯測試2是否成立，若邏輯測試2條件式成立，即取成立值2之運算結果；反之，繼續判斷邏輯測試3是否成立，若邏輯測試3條件式成立，即取成立值3之運算結果……。

若所有條件均不成立，其回應值為#NA!錯誤。

此函數之目的，在縮減原單獨使用IF()函數時，若為多重條件，得組合成很長之巢狀IF()。

求成績等級

例如，擬依右示條件分別給予成績之等級：

成績	等級
90 ～	A
80 ～ 89	B
70 ～ 79	C
60 ～ 69	D
～ 59	F

範例10.xlsx『成績等級-IF』工作表之D2，以單一條件IF()函數進行判斷成績等級，其運算式將為：

=IF(C2>=90,"A",IF(C2>=80,"B",IF(C2>=70,"C",IF(C2>=60,"D","F"))))

D2									
	A	B	C	D	E	F	G	H	
1	姓名	性別	成績	等級					
2	郭源龍	男	91	A					
3	蔡珮珊	女	89	B					
4	鄧智豪	男	78	C					
5	鄭宇廷	男	48	F					
6	陳薇羽	女	65	D					

一連串的IF()，不僅數量多，且左/右括號要完全配對，有時還不太容易！

但若改為使用多重條件IFS()函數（範例10.xlsx『成績等級-IFS』工作表），其運算式將為：

=IFS(C2>=90,"A",C2>=80,"B",C2>=70,"C",C2>=60,"D",C2<60,"F")

D2									
	A	B	C	D	E	F	G	H	
1	姓名	性別	成績	等級					
2	郭源龍	男	91	A					
3	蔡珮珊	女	89	B					
4	鄧智豪	男	78	C					
5	鄭宇廷	男	48	F					
6	陳薇羽	女	65	D					

可少掉好幾個IF，運算式明顯較短。

求獎金比例

假定，員工之業績獎金係依其業績高低，給予不同之比例：

業績	獎金比例
0~299,999	0.0%
300000~499,999	0.3%
500000~999,999	0.5%
1000000~1,499,999	0.8%

業績	獎金比例
1500000~1,999,999	1.0%
2000000~2,999,999	2.0%
3000000~	3.0%

範例『Fun10-參照.xlsx\獎金比例-IF』之C2，以單一條件IF()函數進行判斷獎金比例，其運算式將為：

```
=IF(B2>=3000000,3%,IF(B2>=2000000,2%,IF(B2>=1500000,1%,IF(B2>=1000000,0.8%,IF(B2>=500000,0.5%,IF(B2>=300000,0.3%,0%)))))
```

C2								
	A	B	C	D	E	F	G	H
1	姓名	業績	獎金比例	業績獎金				
2	吳景新	2,580,000	2.0%	51,600				
3	林書宏	1,025,000	0.8%	8,200				
4	林淑芬	250,000	0.0%	0				
5	蔡桂芳	2,250,000	2.0%	45,000				
6	梁國正	1,380,000	0.8%	11,040				
7	楊佳偉	568,000	0.5%	2,840				
8	黃光輝	3,500,000	3.0%	105,000				

運算式很長，想取得正確結果，還真不太容易！

但若改為使用多重條件IFS()函數（範例10.xlsx『獎金比例-IFS』工作表），其運算式將為：

```
=IFS(B2>=3000000,3%,B2>=2000000,2%,B2>=1500000,1%,B2>=1000000,0.8%,B2>=500000,0.5%,B2>=300000,0.3%,B2>0,0%)
```

C2								
	A	B	C	D	E	F	G	H
1	姓名	業績	獎金比例	業績獎金				
2	吳景新	2,580,000	2.0%	51,600				
3	林書宏	1,025,000	0.8%	8,200				
4	林淑芬	250,000	0.0%	0				
5	蔡桂芳	2,250,000	2.0%	45,000				
6	梁國正	1,380,000	0.8%	11,040				
7	楊佳偉	568,000	0.5%	2,840				
8	黃光輝	3,500,000	3.0%	105,000				

運算式會短一點點，但其實還是蠻長的！若能改為下節之VLOOKUP()函數，將會更便捷許多。

10-2

垂直查表VLOOKUP()

VLOOKUP(查表依據,表格,第幾欄,[是否不用找到完全相同值])

VLOOKUP(lookup_value,table_array,col_index_num,[range_lookup])

在一表格的最左欄中，尋找含查表依據的欄位，並傳回同一列中第幾欄所指定之儲存格內容。式中，方括號所包圍之內容，表該部份可省略。

表格是要在其中進行找尋資料的陣列範圍，且必須按其第一欄之內容遞增排序。

[是否不用找到完全相同值]為一邏輯值，為TRUE（或省略）時，如果找不到完全符合的值，會找出僅次於查表依據的值。當此引數值為FALSE時，必須找尋完全符合的值，如果找不到，則傳回錯誤值#N/A。

不用找到完全相同值之實例

假定，員工之業績獎金係依其業績高低，給予不同之比例：

業績	獎金比例
0~299,999	0.0%
300000~499,999	0.3%
500000~999,999	0.5%
1000000~1,499,999	0.8%
1500000~1,999,999	1.0%
2000000~2,999,999	2.0%
3000000~	3.0%

茲將其對照表安排於範例『Fun10-參照.xlsx\ VLOOKUP1』之A3:B9：

	A	B
1	業績與獎金比例對照表	
2	業績	獎金比例
3	0	0.0%
4	300,000	0.3%
5	500,000	0.5%
6	1,000,000	0.8%
7	1,500,000	1.0%
8	2,000,000	2.0%
9	3,000,000	3.0%

安排此一表格時，**標題之文字內容並無作用**，重點為代表業績及獎金比例之數字，**第一個0很重要**，很多使用者直接於0的位置上輸入300,000，將會使業績未滿300,000者，找不到可用之獎金比例，而顯示錯誤值#N/A。此外，**務必記得要依第一欄之業績內容遞增排序**。

假定，各員工之基本薪及業績資料為：

	A	B	C	D	E	F
12	員工編號	姓名	基本薪	業績	業績獎金	總所得
13	1001	吳景新	25,000	2,580,000		
14	1002	林書宏	28,000	1,025,000		
15	1003	林淑芬	30,000	250,000		
16	1004	蔡桂芳	35,000	2,250,000		
17	1005	梁國正	28,000	1,380,000		
18	1006	楊佳偉	40,000	568,000		
19	1007	黃光輝	40,000	3,500,000		

於E欄，擬依D欄之業績計算其業績獎金。首先，於E13處可使用

=VLOOKUP(D13,\$A\$3:\$B\$9,2,TRUE)

依D欄之業績（**查表依據**），於A3:B9（**表格**）中找出適當（**第2欄**）之獎金百分比：

E13	:	x	✓	f	=VLOOKUP(D13,\$A\$3:\$B\$9,2,TRUE)	
	A	B	C	D	E	F
12	員工編號	姓名	基本薪	業績	業績獎金	總所得
13	1001	吳景新	25,000	2,580,000	0.02	
14	1002	林書宏	28,000	1,025,000	0.008	
15	1003	林淑芬	30,000	250,000	0	
16	1004	蔡桂芳	35,000	2,250,000	0.02	
17	1005	梁國正	28,000	1,380,000	0.008	
18	1006	楊佳偉	40,000	568,000	0.005	
19	1007	黃光輝	40,000	3,500,000	0.03	

最後一個引數為何要使用TRUE？這是因為業績內容很少恰好等於A3:A9的間距數字。將其安排為TRUE（或省略）時，於A3:A9找不到完全符合D欄之業績值，將找出僅次於查表依據的值。如：業績1,025,000者，不可能會給予與1,500,000同列之1%為獎金比例，而是找到僅次於1,025,000之1,000,000，而回應與1,000,000同列之0.8%為其獎金比例。

此外，安排業績與其獎金比例之表格原範圍為A3:B9，為了方便向下抄給其它儲存格，應記得將其安排為\$A\$3:\$B\$9。

於判斷查表所取得之獎金比例無誤之後，將其乘上業績：

=VLOOKUP(D13,\$A\$3:\$B\$9,2,TRUE)*D13

即可算出業績獎金：

E13							
12	員工編號	姓名	基本薪	業績	業績獎金	總所得	
13	1001	吳景新	25,000	2,580,000	51,600		
14	1002	林書宏	28,000	1,025,000	8,200		
15	1003	林淑芬	30,000	250,000	-		
16	1004	蔡桂芳	35,000	2,250,000	45,000		
17	1005	梁國正	28,000	1,380,000	11,040		
18	1006	楊佳偉	40,000	568,000	2,840		
19	1007	黃光輝	40,000	3,500,000	105,000		

最後，將C欄之基本薪加上E欄業績獎金，即可獲致F欄之總所得：

F13							
12	員工編號	姓名	基本薪	業績	業績獎金	總所得	
13	1001	吳景新	25,000	2,580,000	51,600	76,600	
14	1002	林書宏	28,000	1,025,000	8,200	36,200	
15	1003	林淑芬	30,000	250,000	-	30,000	
16	1004	蔡桂芳	35,000	2,250,000	45,000	80,000	
17	1005	梁國正	28,000	1,380,000	11,040	39,040	
18	1006	楊佳偉	40,000	568,000	2,840	42,840	
19	1007	黃光輝	40,000	3,500,000	105,000	145,000	



續前例，假定所得稅率為：

所得	稅率
0~30,000	0.0%
30,001~50,000	3.0%
50,001~80,000	4.5%
80,001~100,000	8.0%
100,001~150,000	10.0%
150,001~200,000	16.0%
200,001~	20.0%

試依查表取得適當稅率計算所得稅，並計算扣除所得稅後之淨所得：（範例『Fun10-參照.xlsx\LOOKUP-淨所得』）

	B	C	D	E	F	G	H
12	姓名	基本薪	業績	業績獎金	總所得	所得稅	淨所得
13	吳景新	25,000	2,580,000	51,600	76,600	3,447	73,153
14	林書宏	28,000	1,025,000	8,200	36,200	1,086	35,114
15	林淑芬	30,000	250,000	-	30,000	-	30,000
16	蔡桂芳	35,000	2,250,000	45,000	80,000	3,600	76,400
17	梁國正	28,000	1,380,000	11,040	39,040	1,171	37,869
18	楊佳偉	40,000	568,000	2,840	42,840	1,285	41,555
19	黃光輝	40,000	3,500,000	105,000	145,000	14,500	130,500

必須找到完全相同值之實例

前例之VLOOKUP()中的最後一個引數使用TRUE，如果找不到完全符合的值，會找出僅次於查表依據的值。但，於範例『Fun10-參照.xlsx\LOOKUP2』中：

	A	B	C	D	E	F	G	H
1	編號	姓名	性別	部門	職稱	生日	地址	電話
2	1201	張惠真	女	會計	主任	1981/11/27	台北市民生東路三段68號六樓	(02)2517-6399
3	1203	呂姿瑩	女	人事	主任	1988/02/27	台北市興安街一段15號四樓	(02)2515-5428
4	1208	吳志明	男	業務	主任	1977/06/01	台北市內湖路三段148號二樓	(02)2517-6408
5	1218	黃啟川	男	業務	專員	1992/05/09	台北市合江街124號五樓	(02)2736-3972
6	1220	謝龍盛	男	業務	專員	1987/04/29	桃園市成功路338號四樓	(03)8894-5677
7	1316	孫國寧	女	門市	主任	1984/08/21	台北市北投中央路12號三樓	(02)5897-4651
8	1318	楊桂芬	女	門市	銷售員	1982/08/23	台北市龍江街23號三樓	(02)2555-7892
9	1440	梁國棟	男	業務	專員	1992/12/06	台北市敦化南路138號二樓	(02)7639-8751
10	1452	林美惠	女	會計	專員	1975/07/26	基隆市中正路二段12號二樓	(03)3399-5146

雖同樣以數字性質之編號進行找尋，就不可以於找不到完全符合的編號值，即以編號較小的另一筆記錄內容來替代。故應將VLOOKUP()中的最後一個引數，改為使用FALSE，必須要找尋完全符合的值，如果找不到，則傳回錯誤值#N/A。


假定，要利用使用者所輸入之員工編號，傳回如下示之表格內容：

	A	B	C	D	E
12					
13		編號	1440	姓名	梁國棟
14		性別	男	部門	業務
15		職稱	專員	生日	1992/12/06
16		地址	台北市敦化南路138號二樓		
17		電話	(02)7639-8751		

其處理步驟為：

1 安排妥表格外觀

	A	B	C	D	E
12					
13		編號		姓名	
14		性別		部門	
15		職稱		生日	
16		地址			
17		電話			

其中，C16:E16與C17:E17係分別於選取後，以『常用/對齊方式/跨欄置中』鈕右側之向下箭頭，續選「合併儲存格(M)」，將其設定為合併儲存格。

2 於C13輸入一已存在之員工編號（如：1440）

3 於E13輸入

```
=VLOOKUP($C$13,$A$2:$H$10,2,FALSE)
```

公式，可找出該編號所對應之員工姓名（第2欄）：

E13							
	A	B	C	D	E	F	G
12							
13		編號	1440	姓名	梁國棟		
14		性別		部門			
15		職稱		生日			
16		地址					
17		電話					



前兩個引數，使用含\$之絕對參照，係因此公式仍要抄給其它儲存格使用。最後一個引數，使用FALSE，表一定要找到完全相同之員

工編號；否則，即顯示#N/A之錯誤，而不是找一個編號較低者來替代。

4 按『常用/剪貼簿/複製』鈕，記下E13之內容

5 按住 **Ctrl** 鍵，選取C14:C15、E14:E15與C16:C17等儲存格

	A	B	C	D	E
12					
13		編號	1440	姓名	梁國棟
14		性別		部門	
15		職稱		生日	
16		地址			
17		電話			

6 按『常用/剪貼簿/貼上』之下拉鈕，選按『貼上/公式(F)』鈕。可獲致

C16					=VLOOKUP(\$C\$13,\$A\$2:\$H\$10,2,FALSE)	
	A	B	C	D	E	F
12						
13		編號	1440	姓名	梁國棟	
14		性別	梁國棟	部門	梁國棟	
15		職稱	梁國棟	生日	梁國棟	
16		地址	梁國棟			
17		電話	梁國棟			

7 將C14:C15、E14:E15與C16:C17等儲存格之公式內容的第三個引數，由2分別改為所對應之欄數。如：

```
C14 =VLOOKUP($C$13,$A$2:$H$10,3,FALSE)
E14 =VLOOKUP($C$13,$A$2:$H$10,4,FALSE)
C15 =VLOOKUP($C$13,$A$2:$H$10,5,FALSE)
E15 =VLOOKUP($C$13,$A$2:$H$10,6,FALSE)
C16 =VLOOKUP($C$13,$A$2:$H$10,7,FALSE)
C17 =VLOOKUP($C$13,$A$2:$H$10,8,FALSE)
```

可獲致

C17					=VLOOKUP(\$C\$13,\$A\$2:\$H\$10,8,FALSE)	
	A	B	C	D	E	F
12						
13		編號	1440	姓名	梁國棟	
14		性別	男	部門	業務	
15		職稱	專員	生日	1992/12/06	
16		地址	台北市敦化南路138號二樓			
17		電話	(02)7639-8751			

往後，於C13處輸入員工編號，即可取得其相關之所有資料內容：

	A	B	C	D	E
12					
13		編號	1318	姓名	楊桂芬
14		性別	女	部門	門市
15		職稱	銷售員	生日	1982/08/23
16		地址	台北市龍江街23號三樓		
17		電話	(02)2555-7892		

但若輸入一個不存在之員工編號（如：1215），即顯示#N/A之錯誤，而不是找一個編號較低者（1208）來替代：

	A	B	C	D	E
12					
13		編號	1215	姓名	#N/A
14		性別	#N/A	部門	#N/A
15		職稱	#N/A	生日	#N/A
16		地址	#N/A		
17		電話	#N/A		

將#N/A改為"找不到"

若要將#N/A改為"找不到"，可使用如

```
=IF(ISNA(VLOOKUP($C$13,$A$2:$H$10,2,FALSE)),"找不到",VLOOKUP($C$13,$A$2:$H$10,2,FALSE))
```

之公式來判斷，當員工編號不存在，即將#N/A改為"找不到"。將所有使用VLOOKUP()之儲存格，均改為類似之公式後，可獲致：（範例『Fun10-參照.xlsx\VLOOKUP2-1』）

E13							
	A	B	C	D	E	F	G
12							
13		編號	1450	姓名	找不到		
14		性別	找不到	部門	找不到		
15		職稱	找不到	生日	找不到		
16		地址	找不到				
17		電話	找不到				

此處之運算式相當冗長，利用IFNA()函數則可簡化其算式，其語法為：

```
IFNA(值,成立時的值)
IFNA(value,value_if_na)
```

當判斷出**值**為#NA!之錯誤，即賦予**成立時的值**；否則，即取用**值**之內容。
故而，E13 運算式將可簡化為：

```
=IFNA(VLOOKUP($C$13,$A$2:$H$10,2,FALSE),"找不到")
```

將所有使用 VLOOKUP() 之儲存格，均改為類似之公式後：

```
C14 =IFNA(VLOOKUP($C$13,$A$2:$H$10,3,FALSE),"找不到")
E14 =IFNA(VLOOKUP($C$13,$A$2:$H$10,4,FALSE),"找不到")
C15 =IFNA(VLOOKUP($C$13,$A$2:$H$10,5,FALSE),"找不到")
E15 =IFNA(VLOOKUP($C$13,$A$2:$H$10,6,FALSE),"找不到")
C16 =IFNA(VLOOKUP($C$13,$A$2:$H$10,7,FALSE),"找不到")
C17 =IFNA(VLOOKUP($C$13,$A$2:$H$10,8,FALSE),"找不到")
```

若 C13 之編號可順利找到，同樣可顯示其記錄內容：

E13							
	A	B	C	D	E	F	G
12							
13		編號	1318	姓名	楊桂芬		
14		性別	女	部門	門市		
15		職稱	銷售員	生日	1982/8/23		
16		地址	台北市龍江街23號三樓				
17		電話	(02)2555-7892				

若 C13 之編號找不到，也不會出現 #NA 之錯誤，而是顯示 "找不到"：


E13							
	A	B	C	D	E	F	G
12							
13		編號	1520	姓名	找不到		
14		性別	找不到	部門	找不到		
15		職稱	找不到	生日	找不到		
16		地址	找不到				
17		電話	找不到				

設定僅能輸入編號進行查詢


由於前例係於 C13 處輸入員工編號進行查詢，但使用者仍可能於 C13 以外的其它儲存格輸入內容。如此，難保不會破壞查詢表中之公式內容。因為，其公式確實複雜，要重打得浪費不少時間！


此時，可以下列步驟，將其設定為僅能於C13輸入編號進行查詢，於其它位置輸入任何資料（或編輯/刪除）均不被允許：


- 1 停於C13，於其上單按滑鼠右鍵，續選「儲存格格式(F)…」，選取『保護』標籤
- 2 取消「鎖定(L)」與「隱藏(I)」




- 3 按  鈕
- 4 按『校閱/變更/保護工作表』 鈕



- 5 輸入密碼，續按  鈕



- 6 再輸入一次完全相同之密碼。按 **確定** 鈕，即可完成設定。如此，僅能於C13輸入編號進行查詢，於其它位置輸入任何資料、編輯或刪除儲存格內容，均將獲致錯誤訊息：



文字串之實例

假定，某公司之產品編號、品名及單價，如範例『Fun10-參照.xlsx\LOOKUP4』之A1:C8所示：

	A	B	C
1	編號	品名	單價
2	A01	電視	23,680
3	A02	冰箱	36,500
4	A03	電腦	28,750
5	B01	電話	1,250
6	B04	答錄機	860
7	C02	隨身碟	420
8	C05	滑鼠	680

建立表格時，必須按A欄之編號遞增排序，但仍允許跳號。

於交易發生時，為方便輸入資料，可於輸入產品編號後，以VLOOKUP()查得其品名及單價。因為，不可能會依編號順序發生交易，故下表並無必須按編號遞增排序之要求，且允許重複出現：

	A	B	C	D	E	F
11	日期	編號	品名	單價	數量	金額
12	2019/8/12	C05				
13	2019/8/12	A01				
14	2019/8/12	A03				
15	2019/8/12	B04				

要利用VLOOKUP()依編號查表取其品名及單價，可先於C12輸入

```
=VLOOKUP($B12,$A$2:$C$8,2,FALSE)
```

可取得品名：

C12		:	x	✓	f	=VLOOKUP(\$B12,\$A\$2:\$C\$8,2,FALSE)	
	A	B	C	D	E	F	G
11	日期	編號	品名	單價	數量	金額	
12	2019/8/12	C05	滑鼠				

由於這也是一個必須要找到完全相同編號的例子，故最後一個引數安排為 FALSE。將其抄給 D12 後，可獲得一完全相同之公式，將其第三個引數改為 3：

`=VLOOKUP($B12,$A$2:$C$8,3,FALSE)`

即可獲得其單價：

D12	:	x	✓	f _x	=VLOOKUP(\$B12,\$A\$2:\$C\$8,3,FALSE)		
	A	B	C	D	E	F	G
11	日期	編號	品名	單價	數量	金額	
12	2019/8/12	C05	滑鼠	680			

將 C12:D12 抄給 C13:D15，即可取得各筆交易之品名及單價：

C12	:	x	✓	f _x	=VLOOKUP(\$B12,\$A\$2:\$C\$8,2,FALSE)		
	A	B	C	D	E	F	G
11	日期	編號	品名	單價	數量	金額	
12	2019/8/12	C05	滑鼠	680			
13	2019/8/12	A01	電視	23680			
14	2019/8/12	A03	電腦	28750			
15	2019/8/12	B04	答錄機	860			

剩下來之工作，僅須輸入各筆交易之數量，即可以單價乘以數量，求得其金額：

F12	:	x	✓	f _x	=D12*E12		
	A	B	C	D	E	F	
11	日期	編號	品名	單價	數量	金額	
12	2019/8/12	C05	滑鼠	680	4	2720	
13	2019/8/12	A01	電視	23680	3	71040	
14	2019/8/12	A03	電腦	28750	5	143750	
15	2019/8/12	B04	答錄機	860	2	1720	

往後，若再有新交易發生，只須繼續向下進行輸入即可，並不用再複製公式，Excel 會自動進行必要之公式的複製。例如，輸入完日期與編號後，即可自動取得品名及單價：

	A	B	C	D	E	F	
11	日期	編號	品名	單價	數量	金額	
12	2019/8/12	C05	滑鼠	680	4	2720	
13	2019/8/12	A01	電視	23680	3	71040	
14	2019/8/12	A03	電腦	28750	5	143750	
15	2019/8/12	B04	答錄機	860	2	1720	
16	2019/8/13	A02	冰箱	36500			

10-4

查表LOOKUP()

向量型

LOOKUP(查表依據,查表向量,結果向量)

LOOKUP(lookup_value,lookup_vector,result_vector)

所使用的兩個向量，均為單列或單欄的陣列。本類型之LOOKUP()函數，會在**查表向量**中找尋**查表依據**之內容，然後移到另一個**結果向量**中的同一個位置上，傳回該儲存格的內容。但應注意：

- 兩向量之儲存格個數應一致
- **查表向量**之內容應事先遞增排序
- 如果於**查表向量**中無法找到**查表依據**之內容，將取用較小的一個值來替代
- 如果**查表依據**之內容小於整個**查表向量**之所有值，將回應#N/A之錯誤值

如，將成績高低分為下列幾組：

成績	組別
0~59	不及格
60~74	中等
75~84	高分
85~100	特優

將其內容安排於範例『Fun10-參照.xlsx\LOOKUP向量』之A1:B5位置。如此，A2:A5即可當查表向量；B2:B5即可當結果向量。

	A	B
1	成績	組別
2	0	不及格
3	60	中等
4	75	高分
5	85	特優

假定，要將成績內容，於其備註欄上填入適當之組別文字，D9處之公式可為：

```
=LOOKUP(C9,$A$2:$A$5,$B$2:$B$5)
```

D9							
	A	B	C	D	E	F	G
8	學號	姓名	成績	備註			
9	1001	李碧莊	78	高分			
10	1002	林淑芬	85	特優			
11	1003	王嘉育	60	中等			
12	1004	吳育仁	82	高分			

由於其結果向量僅能為單列或單欄的陣列。故若假定，要使用學號來找出姓名、成績及備註欄內容。就得標定不同之結果向量，如下表中G10:G12之內容將分別為：

```
G10 =LOOKUP(G9,A9:A17,B9:B17)
```

```
G11 =LOOKUP(G9,A9:A17,C9:C17)
```

```
G12 =LOOKUP(G9,A9:A17,D9:D17)
```

分別使用三組不同的結果向量，才可找到適當之資料內容：

G10						
	E	F	G	H	I	J
8						
9		學號	1003			
10		姓名	王嘉育			
11		成績	60			
12		備註	中等			

陣列型

```
LOOKUP(查表依據,陣列)
```

```
LOOKUP(lookup_value,array)
```

本類型之LOOKUP()函數則會在陣列的第一列（或第一欄），搜尋指定的查表依據，然後傳回其最後一列（或欄）的同一個位置上之儲存格內容。

所以，同上例之要求，要使用此一類型之LOOKUP()函數，依成績高低，於其備註欄上填入適當之組別文字，D9處之公式將改為：(範例『Fun10-參照.xlsx\LOOKUP陣列』)

=LOOKUP(C9,\$A\$2:\$B\$5)

將原分為兩個向量之內容，組合成單一陣列即可：

D9							=LOOKUP(C9,\$A\$2:\$B\$5)
	A	B	C	D	E	F	
1	成績	組別					
2	0	不及格					
3	60	中等					
4	75	高分					
5	85	特優					
6							
7							
8	學號	姓 名	成績	備註			
9	1001	李碧莊	78	高分			
10	1002	林淑芬	85	特優			
11	1003	王嘉育	60	中等			

但由於此類型之LOOKUP()函數，不管陣列之欄列數多寡，將永遠傳回最後一列（或欄）的對應內容。故若要於A9:D17表中，依學號找出姓名、成績及備註欄內容。就得標定不同之三組陣列，分別讓所要的內容安排於最後一欄才可。如下表中G10:G12之內容將分別為：

G10 =LOOKUP(G9,A9:B17)

G11 =LOOKUP(G9,A9:C17)

G12 =LOOKUP(G9,A9:D17)

分別使用三組不同欄數之陣列，每個陣列均讓所要找出之內容安排於最後一欄，才可找到適當之資料內容：

G10						=LOOKUP(G9,A9:B17)
	E	F	G	H	I	
8						
9		學號	1007			
10		姓名	葉婉青			
11		成績	48			
12		備註	不及格			

陣列型

INDEX(陣列,[第幾列],[第幾欄])

INDEX(array,[row_num],[column_num])

本類型之INDEX()函數，可於陣列中找出指定之欄與列交會處之儲存格內容。若該格為空白儲存格，將回應0。

式中，方括號所包圍之內容，表該部份可省略。如果陣列只包含單一的橫列或直欄時，則所對應的[第幾列]或[第幾欄]是可省略的。如果省略了[第幾列]這個引數，則一定要輸入[第幾欄]。如果省略了[第幾欄]這個引數，則一定要輸入[第幾列]。如果陣列含有多列多欄的元素，卻只單獨使用[第幾列]或[第幾欄]，則將以陣列形式傳回陣列中的某一整列或整欄元素。

如，於範例『Fun10-參照.xlsx\INDEX陣列』之功課表中，於B13要找出星期二第三節之科目，可使用

=INDEX(B2:F9,B12,B11)

B13		=INDEX(B2:F9,B12,B11)				
	A	B	C	D	E	F
1	星期 節次	週一	週二	週三	週四	週五
2	1	週會		旅館管理		
3	2	週會	西餐	旅館管理	港式點心	英文
4	3	中餐	西餐	旅館管理	港式點心	英文
5	4	中餐	西餐		港式點心	英文
6	5			通識		
7	6	烘焙		通識	國文	
8	7	烘焙	體育		國文	
9	8	烘焙	體育		國文	
10						
11	星期	2				
12	節次	3				
13	科目	西餐				

本類型之INDEX()函數，若省略標示第幾欄（或第幾列）將傳回該欄（或該列）之全部內容，也就是傳回一個陣列。如，選取範例『Fun10-參照.xlsx\INDEX陣列』之H1:H9，鍵入公式：

=INDEX(A1:F9,,3)

	C	D	E	F	G	H
1	週二	週三	週四	週五		1:F9,,3)
2		旅館管理				
3	西餐	旅館管理	港式點心	英文		
4	西餐	旅館管理	港式點心	英文		
5	西餐		港式點心	英文		
6		通識				
7		通識	國文			
8	體育		國文			
9	體育		國文			

續按 **Ctrl** + **Shift** + **Enter** 完成輸入，將取得整個星期二之功課表：

	C	D	E	F	G	H
1	週二	週三	週四	週五		週二
2		旅館管理				0
3	西餐	旅館管理	港式點心	英文		西餐
4	西餐	旅館管理	港式點心	英文		西餐
5	西餐		港式點心	英文		西餐
6		通識				0
7		通識	國文			0
8	體育		國文			體育
9	體育		國文			體育

由於可傳回一個陣列（相當一個範圍），如果其內為數字，當然也可以拿來進行加總或求極大、極小、……等。

參照型

INDEX(一組或多組範圍,[第幾列],[第幾欄],[第幾組範圍])

INDEX(reference,[row_num],[column_num],[area_num])

本類型之INDEX()函數，可於一組或多組範圍中（相臨或不相臨），指定第幾組範圍中之某一欄列交會處之儲存格內容。當處理者為不相臨之範圍時，得以一對括號將其包圍。

式中，方括號所包圍之內容，表該部份可省略。如果省略了[第幾列]與[第幾欄]，將傳回參照中由[第幾組範圍]指定的區域。如果[第幾組範圍]被省略了，則將使用第一個區域。

如於範例『Fun10-參照.xlsx\INDEX參照』之A1:C8與E1:F7，兩個不等大小之範圍中，要取得第一組範圍A1:C8之第二列第三欄，可使用

```
=INDEX((A1:C8,E1:F7),2,3,1)
```

或

```
=INDEX(A1:C8,2,3)
```

要取得第二組範圍E1:F7之第四列第二欄，可使用

```
=INDEX((A1:C8,E1:F7),4,2,2)
```

或

```
=INDEX(E1:F7,4,2)
```

A14							=INDEX((A1:C8,E1:F7),4,2,2)
	A	B	C	D	E	F	
1	編號	品名	單價		金額	折扣率	
2	A01	電視	23,680		0	0.0%	
3	A02	冰箱	36,500		50,000	2.5%	
4	A03	電腦	28,750		80,000	3.5%	
5	B01	電話	1,250		100,000	5.0%	
6	B04	答錄機	860		150,000	10.0%	
7	C02	隨身碟	1,200		300,000	20.0%	
8	C05	滑鼠	680				
9							
10	第一組之第二列第三欄						
11	23680	← =INDEX((A1:C8,E1:F7),2,3,1)					
12							
13	第二組之第四列第二欄						
14	0.035	← =INDEX((A1:C8,E1:F7),4,2,2)					

10-6

比對MATCH()

MATCH(比對依據,陣列,[比對方式])

MATCH(lookup_value,lookup_array,[match_type])

依指定的[比對方式]，傳回一**陣列**中與**比對依據**內容相符合之相對位置。式中，方括號所包圍之內容，表該部份可省略。比對方式有三種情況：

- 1：陣列內容必須先**遞增**排序，將找到等於或僅次於比對依據的值。省略[比對方式]，其值預設為1。
- 0：陣列內容**不必排序**，將找到完全符合比對依據的值；否則，即顯示#N/A之錯誤值。
- -1：陣列內容必須先**遞減**排序，將找到等於或大於比對依據的值。

如於範例『Fun10- 參照.xlsx\MATCH』表中：

	A	B	C	D	E	F	G
1	票價	基隆	台北	新竹	台中	台南	高雄
2	基隆	0	90	240	500	700	900
3	台北	90	0	200	400	600	800
4	新竹	240	200	0	300	500	720
5	台中	500	400	300	0	300	560
6	台南	700	600	500	300	0	240
7	高雄	900	800	720	560	240	0

想查出任兩站間之票價，無論以HLOOKUP()、VLOOKUP()或LOOKUP()，均得知道其列數與欄數。若用人去算，那就沒啥學問了！

此時，即可使用MATCH()來算出某站名究竟排於第幾欄（或列）。以B9所輸入之"台北"，要判斷出其排列於B1:G1範圍之第幾欄？由於各站名並未事先排妥遞增或遞減順序，故本例之比對方式為0，於E9可使用

=MATCH(B9,A2:A7,0)

來取得2，表"台北"係排列於A2:A7範圍之第2列。同理，以B10所輸入之"高雄"，於E10可使用

```
=MATCH(B10,B1:G1,0)
```

來取得6，表"高雄"係排列於B1:G1範圍之第6欄。故B11即可使用

```
=INDEX(B2:G7,E9,E10)
```

來查得台北到高雄之票價：

B11										=INDEX(B2:G7,E9,E10)
	A	B	C	D	E	F	G	H		
9	起站	台北		列	2	← =MATCH(B9,A2:A7,0)				
10	終站	高雄		欄	6	← =MATCH(B10,B1:G1,0)				
11	票價	800								

若未事先以E9及E10求其欄列數，甚至可直接以

```
=INDEX(B2:G7,MATCH(B9,A2:A7,0),MATCH(B10,B1:G1,0))
```

來查得票價：（範例『Fun10-參照.xlsx\MATCH1』）

B11										=INDEX(B2:G7,MATCH(B9,A2:A7,0), MATCH(B10,B1:G1,0))
	A	B	C	D	E	F	G	H	I	
1	票價	基隆	台北	新竹	台中	台南	高雄			
2	基隆	0	90	240	500	700	900			
3	台北	90	0	200	400	600	800			
4	新竹	240	200	0	300	500	720			
5	台中	500	400	300	0	300	560			
6	台南	700	600	500	300	0	240			
7	高雄	900	800	720	560	240	0			
8										
9	起站	高雄								
10	終站	新竹								
11	票價	720								



使用範例『Fun10-參照.xlsx\MATCH-課表』資料，利用 B11:B12 所輸入之星期幾及第幾節，查出其所對應之課表科目：

	A	B	C	D	E	F
1	星期 節次	週一	週二	週三	週四	週五
2	1	週會		旅館管理		
3	2	週會	西餐	旅館管理	港式點心	英文
4	3	中餐	西餐	旅館管理	港式點心	英文
5	4	中餐	西餐		港式點心	英文
6	5			通識		
7	6	烘焙		通識	國文	
8	7	烘焙	體育		國文	
9	8	烘焙	體育		國文	
10						
11	星期幾	週四				
12	第幾節		3			
13	科目	港式點心				

10-7

選擇 CHOOSE()

CHOOSE(索引值, 結果 1, [結果 2], ...)

CHOOSE(index_num, value1, [value2], ...)

傳回索引值所指之第幾個結果，如：索引值為 2，將傳回結果 2。最多可有 254 組結果。如：

=CHOOSE(2, "優等", "中等", "劣等")

之結果為 "中等"。各不同結果可為數字、文字串、運算公式或參照範圍。如：

=SUM(CHOOSE(3, B2:B4, C2:C4, D2:D4, E2:E4))

將加總第 3 組範圍 D2:D4 之數值。

如果索引值小於 1 或大於結果總數，將傳回錯誤值 #VALUE!。如果索引值不是整數，會將其無條件捨棄到最接近之整數。

求星期幾

由於將任一日期除以7後之餘數（可以MOD求得），若為1表其為星期日、為2表其為星期一、為3表其為星期二、…、為6表其為星期五、為0表其為星期六。故對於範例『Fun10-參照.xlsx\CHOOSE1』工作表B1之日期，可用

```
=CHOOSE(MOD(B1,7)+1,"六","日","一","二","三","四","五")
```

來求得其為星期幾：

B2 =CHOOSE(MOD(B1,7)+1,"六","日","一","二","三","四","五")									
	A	B	C	D	E	F	G	H	I
1	日期	2016/6/12	除以7之餘數	1	← =MOD(B1,7)				
2	星期	日							
3									
4		星期日	← =B1						



秘訣

事實上，數值格式已提供有將日期轉為星期幾之格式。

將業績分級

假定，將業績以一百萬分為三級：

- 未滿一百萬：待加強
- 一至二百萬：尚可
- 二百萬以上：優等

以範例『Fun10-參照.xlsx\CHOOSE2』工作表C2之業績言，即可先以

```
MATCH(C2,{0,1000000,2000000},1)
```

求得其業績應歸入第幾類？大括號內包圍之內容表其為一陣列，於此處應將業績依遞增順序排列，最後一個1引數，表依遞增順序比對，若找不到完全相同值，將以僅小於比對值之內容替代。

故於 D2 可使用

```
=CHOOSE(MATCH(C2,{0,1000000,2000000},1),"待加強","尚可","優等")
```

來取得其備註欄之字串：

D2								
	A	B	C	D	E	F	G	H
1	編號	姓名	業績	備註				
2	1201	張惠真	1,362,580	尚可				
3	1203	呂姿瑩	687,200	待加強				
4	1208	吳志明	2,586,950	優等				

由於，本例之分組間距恰為一百萬，故亦可利用將業績除以一百萬後之數字來取得備註資料。不過，為免索引值小於1或大於結果總數，而傳回錯誤值#VALUE!。故將範例『Fun10- 參照.xlsx\CHOOSE3』工作表D2之內容改為：

```
=CHOOSE(IF(INT(C2/1000000)+1>=3,3,INT(C2/1000000)+1),"待加強","尚可","優等")
```

將`INT(C2/1000000)`先加1，就是要避免索引值小於1；若`INT(C2/1000000)+1`之結果超過3，就全改為3，以免索引值大於結果總數。其結果為：

D2								
	A	B	C	D	E	F	G	H
1	編號	姓名	業績	備註				
2	1201	張惠真	1,362,580	尚可				
3	1203	呂姿瑩	687,200	待加強				
4	1208	吳志明	2,586,950	優等				



假定，於範例『Fun10-參照.xlsx\CHOOSE-獎金』工作表之內容中，加入一業績獎金欄，以

未滿一百萬：1.0%

一至二百萬：1.5%

二百萬以上：3.0%

之百分比，計算其業績獎金：

	A	B	C	D	E
1	編號	姓名	業績	業績獎金	備註
2	1201	張惠真	1,362,580	20,439	尚可
3	1203	呂姿瑩	687,200	6,872	待加強
4	1208	吳志明	2,586,950	77,609	優等

依選擇求加總

CHOOSE() 函數之各組結果，也可以是參照位址。如，於範例『Fun10-參照.xlsx\CHOOSE4』工作表B7即可以

```
=SUM(CHOOSE(B6,B2:B4,C2:C4,D2:D4,E2:E4,F2:F4,G2:G4))
```

來決定應加總哪一組範圍：

B7	:	X	✓	fx	=SUM(CHOOSE(B6,B2:B4,C2:C4,D2:D4,E2:E4,F2:F4,G2:G4))				
	A	B	C	D	E	F	G	H	I
1		一月	二月	三月	四月	五月	六月		
2	電視	3,600	4,200	5,500	4,800	4,500	3,800		
3	電冰箱	2,400	2,600	2,550	3,000	3,800	4,000		
4	冷氣機	2,500	2,000	3,650	4,200	6,400	8,000		
5									
6	加總月份	2							
7	合計	8,800							

若範例『Fun10-參照.xlsx\CHOOSE5』工作表B6處並非數字，而是文字的"一月"、"二月"、……。就仍得以

```
MATCH(B6,B1:G1,0)
```

來判斷B6之月份係排於B1:G1之第幾個順位？然後再以

=SUM(CHOOSE(MATCH(B6,B1:G1,0),B2:B4,C2:C4,D2:D4,E2:E4,F2:F4,G2:G4))

來決定應加總哪一組範圍：

B7										
	A	B	C	D	E	F	G	H	I	J
1		一月	二月	三月	四月	五月	六月			
2	電視	3,600	4,200	5,500	4,800	4,500	3,800			
3	電冰箱	2,400	2,600	2,550	3,000	3,800	4,000			
4	冷氣機	2,500	2,000	3,650	4,200	6,400	8,000			
5										
6	加總月份	三月								
7	合計	11,700								



範例『Fun10-參照.xlsx\CHOOSE-業績』工作表，以CHOOSE()完成可依輸入之文字月份，求算各月之極大與極小之金額及其品名：

	A	B	C	D	E	F	G
1		一月	二月	三月	四月	五月	六月
2	電視	3,600	4,200	5,500	4,800	4,500	3,800
3	電冰箱	2,400	2,600	2,550	3,000	3,800	4,000
4	冷氣機	2,500	2,000	3,650	4,200	6,400	8,000
5							
6	加總月份	二月					
7	合計	8,800					
8							
9	極大	4,200	品名	電視			
10	極小	2,000	品名	冷氣機			

提示：D9可為

=CHOOSE(MATCH(B9,CHOOSE(MATCH(B6,B1:G1,0),B2:B4,C2:C4,D2:D4,E2:E4,F2:F4,G2:G4),0),"電視","電冰箱","冷氣機")

或

=INDEX(A2:A4,MATCH(B9,CHOOSE(MATCH(B6,B1:G1,0),B2:B4,C2:C4,D2:D4,E2:E4,F2:F4,G2:G4),0))

10-9

間接參照 INDIRECT()

INDIRECT(文字表示之位址)

INDIRECT(ref_text)


本函數可傳回一文字串所指定的參照位址，再依該參照位址內容進行運算或顯示資料。如：範例『Fun10-參照.xlsx\INDIRECT1』工作表B1內容為"D1:D3"之文字串，則

=SUM(INDIRECT(B1))

即等於是

=SUM(D1:D3)

B4						
	A	B	C	D	E	
1	位址	D1:D3		100		
2				200		
3				250		
4	間接參照	550				

因此，若曾事先選取範例『Fun10-參照.xlsx\INDIRECT2』工作表B1:G4，以『公式/已定義之名稱/從選取範圍建立』 鈕，將各月份之銷售額範圍分別命名為『一月』（B2:B4）、『二月』（C2:C4）、……、『六月』（G2:G4）。於B6輸入某一月份後，即可以

=SUM(INDIRECT(B6))

來取得其範圍名稱，並進行加總：

B7							
	A	B	C	D	E	F	G
1		一月	二月	三月	四月	五月	六月
2	電視	3,600	4,200	5,500	4,800	4,500	3,800
3	電冰箱	2,400	2,600	2,550	3,000	3,800	4,000
4	冷氣機	2,500	2,000	3,650	4,200	6,400	8,000
5							
6	加總月份	六月					
7	合計	15,800					

是否比使用 CHOOSE() 來得簡化些？



使用範例『Fun10-參照.xlsx\INDIRECT-業績』工作表，以 INDIRECT() 完成可依輸入之文字月份，求算各月之極大與極小之金額及其品名：

	A	B	C	D	E	F	G
1		一月	二月	三月	四月	五月	六月
2	電視	3,600	4,200	5,500	4,800	4,500	3,800
3	電冰箱	2,400	2,600	2,550	3,000	3,800	4,000
4	冷氣機	2,500	2,000	3,650	4,200	6,400	8,000
5							
6	加總月份	二月					
7	合計	8,800					
8							
9	極大	4,200	品名	電視			
10	極小	2,000	品名	冷氣機			

10-10

位移 OFFSET()

OFFSET(參照範圍, 位移列數, 位移欄數, [高度], [寬度])

OFFSET(reference, rows, cols, [height], [width])

傳回某一參照範圍，經向下（上）移動位移列數所指定之列，向右（左）移動位移欄數所指定之欄，移動後之新的參照範圍為何？[高度]與[寬度]係用以指定應傳回之參照範圍的列數及欄數；若省略，將與原參照範圍同高度與寬度。如：（範例『Fun10-參照.xlsx\OFFSET1』）

=OFFSET(A1,2,1)

表自 A1 下移 2 列，右移 1 欄，其結果為 B3。其效果相當

=B3

A5					
	A	B	C	D	E
1	100	525	204		
2	120	156	302		
3	200	148	253		
4					
5	148	← =OFFSET(A1,2,1)			

若位移列（欄）數為負值，表向上（左）移動。如：

=OFFSET(C3,-2,-1)

表自 C3 上移 2 列，左移 1 欄，其結果為 B1。其效果相當

=B1

A6					
	A	B	C	D	E
1	100	525	204		
2	120	156	302		
3	200	148	253		
4					
5	148	← =OFFSET(A1,2,1)			
6	525	← =OFFSET(C3,-2,-1)			

當然，也可以將其結果拿來運算，如：


=SUM(OFFSET(A1:A3,0,1))

表自 A1:A3 右移 1 欄，其結果為 B1:B3。其效果相當加總 B1:B3 之內容：

A7						
	A	B	C	D	E	F
1	100	525	204			
2	120	156	302			
3	200	148	253			
4						
5	148	← =OFFSET(A1,2,1)				
6	525	← =OFFSET(C3,-2,-1)				
7	829	← =SUM(OFFSET(A1:A3,0,1))				

10-17

利用資料庫統計函數建資料表

按『**資料/預測/模擬分析**』 鈕，續選「**運算列表(T)...**」，可用以依公式建立單變數或雙變數之假設分析表（what-if table）或交叉分析表（cross-tabulating）。其類型又隨所使用之變數個數而有：


■ 單變數資料表

資料表中，僅使用到單一變數，以分析當此變數內容發生變化時，相關公式之結果將產生何種變化。

■ 雙變數資料表

資料表中，同時使用兩個不同變數，用以分析當此兩個變數內容發生變化時，相關公式之結果將產生何種變化。

單變數資料表

由於，資料庫統計函數中，均含有一準則範圍之引數（argument），若能將變數欄之不同資料代入該準則範圍之條件式位置，即可利用『**資料/預測/模擬分析**』 鈕，續選「**運算列表(T)...**」來建表，以求得各種不同條件情況之統計量。

茲假定，欲於範例『Fun10-參照.xlsx\運算列表1』房屋資料中：

	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	棟數	坪數	狀況	售屋者	建造日期	售價(萬)
2	B03	士林區	德行東路	4	36	極佳	屋主自售	2005/03/16	1020
3	A09	士林區	忠誠路	2	65	極佳	仲介公司	2008/01/13	1800
4	B05	中山區	合江街	4	18	尚可	仲介公司	2002/12/18	980
5	A03	中山區	中山北路	3	28	極佳	仲介公司	1998/02/11	1020
6	B14	中山區	民生東路	1	38	尚可	屋主自售	1994/09/13	1200
7	B07	內湖區	麗山街	3	55	尚可	屋主自售	2007/02/15	1420
8	A01	內湖區	內湖路	3	45	尚可	屋主自售	2000/11/09	1450
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960

求得各不同地區房屋之：戶數、最大坪數、最小坪數與每坪單價等資料。

將其A12:A13安排為條件準則範圍，其A13目前無任何內容，將來可自E13:E16分別取得各地區別以當作過濾條件。另於F12:I12輸入下列運算式：

F12 =DCOUNTA(A1:I9,1,A12:A13)
 G12 =DMAX(A1:I9,5,A12:A13)
 H12 =DMIN(A1:I9,5,A12:A13)
 I12 =DSUM(A1:I9,9,A12:A13)/DSUM(A1:I9,5,A12:A13)

以當作求：房屋戶數、最大坪數、最小坪數與每坪單價等資料之運算公式。如：

F12	=DCOUNTA(A1:I9,1,A12:A13)								
	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:A13					戶數	最大坪數	最小坪數	每坪單價
12	地區					8	65	12	33.16
13					士林區				
14					中山區				
15					內湖區				
16					大同區				

這幾個公式之共同點即以A12:A13為準則範圍，一旦將變數欄（E13:E16）之"士林區"、"中山區"、"內湖區"或"大同區"，代入準則範圍之A13位置，即可分別取得各不同地區別當作過濾條件。

完成事前準備工作後，依下列步驟執行：

- 1 選取E12:I16為建表範圍

- 2 按『資料/預測/模擬分析』 鈕，續選「運算列表(T)…」，轉入『資料表』對話方塊。於『欄變數儲存格(C)：』後單按滑鼠

- 3 續以滑鼠單按A13儲存格

	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:A13					戶數	最大坪數	最小坪數	每坪單價
12	地區					8	65	12	33.16
13					士林區				
14					中山區				
15					內湖區				
16					大同區				
17									
18									
19									


運算列表

列變數儲存格(B):

欄變數儲存格(C):

確定 取消


因F12:I12等所引用之公式格，均以A12:A13為其準則範圍，而其A13需使用到地區別當作過濾條件。目前資料表範圍所定義之變數欄內容乃為各種不同地區別，故於此處應輸入A13，以便將"士林區"、"中山區"、"內湖區"或"大同區"，分別代入A13位置當作準則範圍之過濾條件，以求算各相關統計量。

- 4 按  鈕完成輸入，可一舉求得各不同地區之房屋的：戶數、最大坪數、最小坪數與每坪單價等資料

	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:A13					戶數	最大坪數	最小坪數	每坪單價
12	地區					8	65	12	33.16
13					士林區	2	65	36	27.92
14					中山區	3	38	18	38.10
15					內湖區	2	55	45	28.70
16					大同區	1	12	12	80.00

雙變數資料表

兩個變數均為字串標記

由於，資料庫統計函數中均有一準則範圍之引數，且其準則範圍內亦可使用兩個欄名。若能將欄/列變數之不同字串資料，代入準則範圍內，適當欄名底下之條件式位置，即可利用『[資料/預測/模擬分析](#)』 鈕，續選「[運算列表\(T\)...](#)」來建表，以求得各種不同組合條件之情況下的統計量，構成一交叉分析表。

茲假定，欲於前文房屋資料中，求得各地區中不同房屋狀況的家數。首先，於範例『Fun10-參照.xlsx\運算列表2』工作表E13:E16輸入各地區別資料當作欄變數；另於F12:G12輸入各狀況別資料當作列變數。接著，於E12輸入：

```
=DCOUNTA(A1:I9,1,A12:B13)
```

以當作此交叉表之運算公式。


另將A12:B13安排為準則範圍，準則範圍內含地區與狀況兩個欄名，其A13、B13處目前均無任何內容，將來A13可自欄變數之範圍（E13:E16）分別取得各地區別；而B13則可自列變數之範圍（F12:G12）分別取得各房屋狀況，以作為過濾條件，再依公式求算符合條件之房屋數。如：

E12					=DCOUNTA(A1:I9,1,A12:B13)				
	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:B13							總計	
12	地區	狀況			8	極佳	尚可		
13					士林區				
14					中山區				
15					大同區				
16					內湖區				
17				總計					

接著，依下列步驟執行：

- 1 將建表範圍選取為E12:H17，涵蓋欄變數及列變數之所有內容以及其交會處之公式（為何多選一列及一欄空格？詳下文說明）

E12					=DCOUNTA(A1:I9,1,A12:B13)				
	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:B13							總計	
12	地區	狀況			8	極佳	尚可		
13					士林區				
14					中山區				
15					大同區				
16					內湖區				
17				總計					

- 2 按『資料/預測/模擬分析』鈕，續選「運算列表(T)…」，轉入『資料表』對話方塊

- 3 於『列變數儲存格(R)：』後單按滑鼠，續以滑鼠選按B13儲存格

因公式中，所使用之準則範圍的『狀況』內容將置於B13，而目前列變數之內容為各狀況別資料，故於『列變數儲存格(R)：』後，輸入B13，以利將列變數之內容代入B13位置。

4 於『欄變數儲存格(C)：』後單按滑鼠，續以滑鼠選按 A13 儲存格

由於公式中，所使用之準則範圍的『地區』內容將置於 A13，而目前欄變數之內容為各地區別資料，故於『欄變數儲存格(C)：』後，輸入 A13，以利將欄變數之內容代入 A13 位置。

	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍 A12:B13							總計	
12	地區	狀況			8	極佳	尚可		
13					士林區				
14					中山區				
15					大同區				
16					內湖區				
17				總計					
18									
19									

5 按 **確定** 鈕，完成輸入。獲致建表內容

E12	:	✕	✓	<i>f_x</i>	=DCOUNTA(A1:I9,1,A12:B13)			
	A	B	C	D	E	F	G	H
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11
10								
11	準則範圍A12:B13							總計
12	地區	狀況			8	極佳	尚可	
13					士林區	2	0	2
14					中山區	1	2	3
15					大同區	1	0	1
16					內湖區	0	2	2
17				總計		4	4	8

可分別將地區別代入 A13（當代入 E17 之空白資料，即等於無『地區』條件，可用以求算欄總計）；將狀況別代入 B13（當代入 H12 之空白資料，即等於無『狀況』條件，可用以求算列總計）。利用 E12 之

=DCOUNTA(A1:I9,1,A12:B13)


公式，一舉算出各地區中各種房屋狀況的家數。

您可能會有疑問，為何不將兩個『總計』字串，安排於 E17 及 H12？但這樣會獲致欄/列之總計部份均為 0：

E12									=DCOUNTA(A1:I9,1,A12:B13)
	A	B	C	D	E	F	G	H	
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	
10									
11	準則範圍A12:B13								
12	地區	狀況			8	極佳	尚可	總計	
13					士林區	2	0	0	
14					中山區	1	2	0	
15					大同區	1	0	0	
16					內湖區	0	2	0	
17					總計	0	0	0	

何故？因為，無論是『地區』或『狀況』，均無"總計"字串所致。

使用條件式

資料庫統計函數中之準則範圍內，亦可使用條件式當過濾條件。所以，也可利用按『資料/預測/模擬分析』 鈕，續選「運算列表(T)…」，以求得各種不同條件之情況下的統計量。

茲假定，欲於前文房屋資料中，將坪數以30坪為界分為兩組；另將售價以1200萬為界分為兩組，求算不同坪數與售價之房屋家數。首先，於範例『Fun10-參照.xlsx\運算列表3』工作表E14:E15內輸入<=30與>30當作欄變數；另於F13:G13輸入<=1200與>1200當作列變數。接著，於E13輸入：

```
=DCOUNTA(A1:I9,1,A12:B13)
```

以當作此交叉表之運算公式。

另將A12:B13安排為準則範圍，準則範圍內含『坪數』與『售價(萬)』兩個欄名，其A13、B13處目前均無任何內容，將來A13可自欄變數範圍(E14:E15)，分別取得坪數之比較條件式；而B13則可自列變數範圍(F13:G13)，分別取得售價之比較條件式，以當作過濾條件，再依公式求算符合條件之房屋數。如：

E13									
	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:B13								
12	坪數	售價(萬)				售價	合計		
13					8	<=1200	>1200		
14				坪數	<=30				
15				坪數	>30				
16				合計					

接著，依下列步驟執行：

- 1 將建表範圍選取為E13:H16

- 2 按『資料/預測/模擬分析』鈕，續選「**運算列表(T)...**」，轉入『資料表』對話方塊

- 3 於『列變數儲存格(R)：』後，單按滑鼠，續以滑鼠選按B13儲存格

因公式中，所使用之準則範圍的『售價(萬)』比較式將置於B13，而目前列變數之內容為售價之比較式，故於『列變數儲存格(R)：』後，輸入B13，以利將售價之比較式代入B13位置。

- 4 於『欄變數儲存格(C)：』後，單按滑鼠，續以滑鼠選按A13儲存格

由於公式中，所使用之準則範圍的『坪數』比較式將置於A13，而目前欄變數之內容為坪數之比較式，故於『欄變數儲存格(C)：』後，輸入A13，以利將坪數之比較式代入A13位置。

	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:B13								
12	坪數	售價(萬)				售價	合計		
13					8	<=1200	>1200		
14				坪數	<=30				
15				坪數	>30				
16				合計					
17									
18									
19									
20									

- 5 按  鈕，完成輸入。獲致建表內容

E13	=DCOUNTA(A1:I9,1,A12:B13)								
	A	B	C	D	E	F	G	H	I
1	編號	地區	街道	樓數	坪數	狀況	售屋者	建造日期	售價(萬)
9	A11	大同區	延平北路	4	12	極佳	仲介公司	1990/06/11	960
10									
11	準則範圍A12:B13								
12	坪數	售價(萬)				售價	合計		
13					8	<=1200	>1200		
14				坪數	<=30	3	0		3
15				坪數	>30	2	3		5
16				合計		5	3		8

可分別將坪數比較條件代入A13（當代入E16之空白資料，即等於無坪數條件，可用以求算列總計）；可分別將售價比較條件代入B13（當代入H13之空白資料，即等於無售價條件，可用以求算欄總計）。利用E13之

=DCOUNTA(A1:I9,1,A12:B13)

公式，一舉算出不同坪數與售價的房屋家數。

10-18

取得資料DGET()

DGET(資料庫表單,欄名或第幾欄,準則範圍)

DGET(database,field,criteria)

本函數可於資料庫表單中，依準則範圍所標示之條件，過濾出第一筆符合條件之記錄，並取回其欄名或第幾欄所指定之欄位。

如於範例『Fun10-參照.xlsx\DGET』工作表A1:H10之資料庫表單中：

	A	B	C	D	E	F	G	H
1	編號	姓名	性別	部門	職稱	生日	地址	電話
2	1201	張惠真	女	會計	主任	1980/05/29	台北市民生東路三段68號六樓	(02)2517-6399
3	1203	呂姿瑩	女	人事	主任	1976/08/28	台北市興安街一段15號四樓	(02)2515-5428
4	1208	吳志明	男	業務	主任	1965/12/01	台北市內湖路三段148號二樓	(02)2517-6408
5	1218	黃啟川	男	業務	專員	1980/11/08	台北市合江街124號五樓	(02)2736-3972
6	1220	謝龍盛	男	業務	專員	1975/10/29	桃園市成功路338號四樓	(03)8894-5677
7	1316	孫國寧	女	門市	主任	1973/02/20	台北市北投中央路12號三樓	(02)5897-4651
8	1318	楊桂芬	女	門市	銷售員	1971/02/22	台北市龍江街23號三樓	(02)2555-7892
9	1440	梁國棟	男	業務	專員	1981/06/07	台北市敦化南路138號二樓	(02)7639-8751
10	1452	林美惠	女	會計	專員	1964/01/25	基隆市中正路二段12號二樓	(03)3399-5146
11								
12		編號	性別	姓名				
13		1318						
14		部門	職稱	生日				
15								
16		電話	地址					
17								

以B12:B13為準則範圍，要取得之欄名或第幾欄可直接使用C12、D12、B14、C14、D14、B16與C16之標題字。故將其相關儲存格安排成下示之公式：

```
C13  =DGET($A$1:$H$10,C12,$B$12:$B$13)
D13  =DGET($A$1:$H$10,D12,$B$12:$B$13)
B15  =DGET($A$1:$H$10,B14,$B$12:$B$13)
C15  =DGET($A$1:$H$10,C14,$B$12:$B$13)
D15  =DGET($A$1:$H$10,D14,$B$12:$B$13)
B17  =DGET($A$1:$H$10,B16,$B$12:$B$13)
C17  =DGET($A$1:$H$10,C16,$B$12:$B$13)
```

即可取得符合條件之指定欄位：

C13	:	X	✓	fx	=DGET(\$A\$1:\$H\$10,C12,\$B\$12:\$B\$13)		
	A	B	C	D	E	F	
11							
12		編號	性別	姓名			
13		1318	女	楊桂芬			
14		部門	職稱	生日			
15		門市	銷售員	1971/2/22			
16		電話	地址				
17		(02)2555-7892	台北市龍江街23號三樓				

往後，只須於B13輸入要查詢之員工編號，即可查得其所有資料。